

第4章 入力・出力仕様と全体の解析処理

ポイント:平面骨組解析の入力仕様 解析全体の流れを理解する

これまでで、平面骨組を解析するための理論的な部分については、既 に学んだ。本章では、骨組解析に必要となる最低限の入力データとその 仕様について説明する。このプログラムは、ExcelのVBAを使用するた め、入力仕様は、全てExcelのセル上で設計され、そこからデータを取 得する。その手続きを含めて、ここでは、解析全体の流れを説明し、ど のように平面骨組の応力解析が実施されるかを学習する。

キーワード 平面骨組の入力仕様 出力仕様 骨組解析の処理全体の流れ

4.2 入力仕様

応力解析を実施するためには、当然、各種の情報が必要となる。ここでは、平面応力解析の入力データとして、最低限必要となる情報を分類し、図 4-1 のように Excel シート上に入力仕様として設計する。まず、入力データを次のように分類する。なお、入力は「データ入力」シートとする。また、出力は、「解析結果」シートとする。

■ ファイル(E)	編集(E) 表示(W) :	挿入印 書式	t(_) ツール(D データ(D)	ウルド	¢₩0	ヘルプモ	þ														
AG2	-	•																				
A 20/101-348	B C D E	F C	N I	3 K	L	H	н	0	p	0	8046	8	Ŧ	ų	¥	VADE	×	Y	z.	AA AB 824060	AC	A
2 101	1000 ABB	1/2/8 11 450	****	10 2000 1		****	ヤン 5倍度 20501	100 R	000 1 201-327 F 2007-4		1001	-39	39. 1	alia .		1011	-39	2519	dir .	821514	R##	
4 <u>8218</u> 5 <u>888</u>	4 2 9	H 191 H 191	2 2	5		- 1	21511	0.82	22644		-	-	- (-						1	- 10	
6 地界发展数 7 电代表重点		6 6	4 8	-		4	_	_														_
1 100000		1 1	2			- 2																
11	9		3		1 1	-	-				=	Ħ									\square	
10	11		11		1 E	11							_									
15 #58594	14		16		1 6	10																
17 18 (1.5.64)(1.64)	16		16		1 1	15																_
19 20 日白水街	3 5		12			12																
21 2/1-4/5-4/2-94t 22 L/0/6H45	6 19 6 20		10		1 1	24	_	_					_									_
21 462.00	-					-	_	_			-		_							-		_
												-	-				-		\rightarrow			-

- 1)コントロールデータ
- 2)解析用データ
 - 2.1 節点座標
 - 2.2 部材の要素情報と節点との結合情報

図 4-1 Excel シート上の入力域の設計

2.3	部材の断面特性
-----	---------

- 2.4 境界条件
- 2.5 荷重項

ここでも、図 4-2 に示す簡単な骨組を使用して、各入 力項目の説明を行う。最初に、入力情報の取得や、解析 に必要となるコントロールデータについて説明する。図 4-3 には、Excel シート上に設計されたコントロールデー タの一覧が示されている。同図から分かるように、この 入力域には解析に必要となる制御用情報が並べられてい る。



図 4-2 例題用骨組

1:節点数
2:部材数
3:要素数
4:境界節点数
5:荷重1(等分布荷重が加わっている部材数)
6:荷重2(節点集中荷重が加わっている節点数)

図 4-3 の入力域では、 例題の解析モデルに関す るデータがセットされて いる。

次に、解析用データに ついて説明する。これら は、コントロールデータ を利用して、セルからプ ログラム内にデータを取 得することになる。

最初は、骨組全体の節 点に関する座標データで ある。図 4-3 では、節点 番号と共に、全体座標系 における節点の座標がセ

🔀 Mi	icrosoft Excel -	平面骨組解	祈.xl	s								
D	🛎 🖪 🔒 🖨 [à 🌾 🐰	Ē	🛍 🝼	∽ - ⊂ - 🦓	Σ f _* ੈ	Z↓					
1817	図 ファイル(E) 編集(E) 表示(V) 挿入① 書式(Q) ツール(T) データ(D) ウィンドウ(W) ^											
2	12 12 13 12 -											
	J7 🔽	· =										
	A B			D	E	F	G					
1	コントロール情報	軗		節点座標								
2					×座標	y座標						
3	節点数	5		1	-300	450						
4	部材数	4		2	-300	150						
5	要素数	2		3	0	0						
6	境界節点数	2		4	300	150						
7	部材荷重数	0		5	300	450						
8	節点荷重数	1		6								
9				7								
10				0								

図 4-3 コントロールデータと節点座標の入力域

ットされている。解析モデルを参照しながら、座標情報を確認されたい。

次に、図 4-4 では、部材に関するデータが並べられており、まず、部 材両端に接合する節点番号とその部材の要素番号である。例題の解析モ デルでは、部材数は4 であり、各部材両端の節点番号が記されている。 このとき、自動的に最初の項が / 端となり、次の項が / 端となる。最後 の項には、当該部材が使用する要素の番号が示されている。

5	Σ	f _*	₿↓	Z↓	10	₽	100%	•	2	Ŧ	8	Ş	» •	Ē₫ "
(<u>D</u>)) (ועזל	~ന്⊘	Ø A	ルプ(Ð	Adobe	PDF(<u>B</u>)					8 ×

F	F G		I	J	К	L	-
		部材の 端部	₿節点番号	-			
y座標			词编	j端	要素番号		ī.
450		1	1	2	1		
150		2	2	3	2		
0		3	3	4	2		
150		4	4	5	1		
450		5					
		6					
		7					
		8					
		9					

図 4-4 部材の両端節点番号と使用要素番号表

図 4-5 は断面特性を 表す要素データである。 ここでは、平面骨組の 総形応力解析として、 最低限のデータ入力域 が設定されている。要 素番号に続いて、ヤン グ係数、断面積、断面 二次モーメントである。 単位は全て、長さは cm で、力は kN とする。

次に、図 4-6 のよう に節点の境界条件を設 定する。条件記載のル ールとして、0 は拘束、 1 は自由とする。例え ば、固定支持の場合は、 0,0,0 であり、ピン支

L	M	N	0	P	Q
	要素番号	ヤング係数	断面積	断面二次モーメント	
	1	20500	90.24	5337.4	
	2	20500	81.92	22964.9	
	3				
	4				
	5				
	6				
	7				
	8				

図 4-5 断面特性を設定する要素データ表

	V21	▼ =				
	Q	R	S	Т	U	v 7
1		境界条件				
2		節点番号	×方向	y方向	回転角	
3		1	0	0	0	
4		4	0	0	0	
5						
6						
7						

図 4-6 骨組境界条件の設定表

持の場合は、0,0,1である。また、水平方 向ローラー支持の場合は、0.1.1となる。 最後の入力項目は荷重項である。ここで は、荷重項は、部材荷重としての等分布荷 重と、節点に加わる集中荷重の2種とする。 節点集中荷重は、荷重の加わっている節 点番号と 2 方向荷重と曲げモーメント荷 重である。図 4-7 には、節点集中荷重の例 が示されている。最後に、部材荷重として、

🔀 Mi	crosoft Excel	- 平面骨組	解析.xls			
] 🖳 🗵	ワァイル(E) 編集(E)表示(⊻);	挿入① 書式	t© ツール①) データ(<u>D</u>)	Ċ
] 🗅 🕯	2 🖬 🔒 🔮	s 🗟 🚏 👌	K 🖻 🛍 :	💅 🗠 - C	× - 🍓 :	Σ
🔁	2 🔁	🖡 🖕 地	キュリティ 🕴	2 🛠 🔟	× •	
	B42	▼ =	-			
	V	W	Х	Y	Z	AA
1		節点荷重				
2		節点番号	×方向	y方向	曲げ	
3		5	0	100	0	
4						
5						
6						
7				l		

) ウィンドウ(W) ヘルプ(<u>H</u>) Adobe PDF(<u>B</u>) $\Sigma f_{\ast} \ge I \xrightarrow{Z} I \xrightarrow{Z} I$ AA AE AB AC 部材荷重(等分布荷重) 部材番号 荷重値 2 10 З 10

等分布荷

図 4-7 節点集中荷重表

重の入力域が図 4-8 に示されている。 同図では、等分布荷重が加わる部材番 号と等分布荷重の値が入力情報となっ ている。この荷重は、材軸方向に直交 する方向に働き、部材座標系で裁荷さ れる。そのため、基本応力を計算した 後、全体座標系に変換し、荷重ベクト ルに組み込む必要がある。

図 4-8 等分布部材荷重表

以上が、この平面骨組応力解析に必 要となる入力情報であり、Excel のシート上に入力領域としてデザイン されている。

本節では、応力解析した結果の出力仕様について説明する。本プログ

4.3 解析結果の出

力仕様

ラムでは、最小限の出力をサポートする。 ここでは、「解析結果」シートに、2種の 解析結果を出力する。一つは、全体座標 系での節点変位であり、3自由度方向(x 方向、 y 方向、回転角)の変位である。 節点番号順に、図 4-9 に示されるように 整理されて出力される。

次に、部材座標系で表された部材の断 面力を表示する。図 4-10 に示されるよ うに、部材番号順にその部材の / 端の断 面力と / 端の断面力、及び、中央の曲げ モーメントが表示される。

🔀 Mi	🔀 Microsoft Excel - 平面骨組解析.xls										
図 ファイル(E) 編集(E) 表示(V) 挿入(D) 書式(Q) ツール(T) データ(D) ウィ											
] D 🛎 🖬 🔒 🎒 🔍 💖 🗼 🛍 🛍 🚿 Ιο - ο - I 🍓 Σ ,											
🔁	🔁 🛃 🐌 🔹 セキュリティ 🐔 🛠 🕍 🛷 🗸										
	C2	▼ :	-								
	A	В	С	D	E						
1											
2											
3	節点番号	u	V	θ							
4	1	0.000	0.000	0.0000							
5	2	6.235	0.495	0.0592							
6	3	-6.235	0.495	-0.0592							
7	4	0.000	0.000	0.0000							
8	5	0.000	13.867	0.0000							
9											
10											
11											
12											

図 4-9 全体座標系で表された節点変位

_									_				
xc	cel - 平面骨組解析.xls												
漏淇	集(E) 表示(V) 挿入(D) 書式(O) ツール(T) データ(D) ウインドウ(W) ヘルプ(H) Adobe PDF(B)												
) (🚑 🗟 ♥ ¾ 🖻 🛍 ダ ∽ - ∝ - 🍓 Σ ≉ ફੈ↓ ¾ 🛍 🦑 100% - ♀ ② ↓ 🕍 🗃 🖓 🗵 🗔 🗆 ⊙ Ў В ≡ 🛱												
	・ セキュリティ… 2 ※ M の 、												
	F	G	Н	I	J	К	L	М					
_			i端			j 端		中央					
	部材番号	N	Q	М	N	Q	М	М					
	1	-3050.000	734.891	88649.387	3050.000	-734.891	131818.059	-21584.336					
	2	-2021.308	954.752	-131818.059	2021.308	2399.349	-110448.218	129940.080					
_	3	-2021.308	2399.349	110448.218	2021.308	954.752	131818.059	129940.080					
_	4	-3050.000	-734.891	-131818.059	3050.000	734.891	-88649.387	-21584.336					
			!				1	I					

図 4-10 解析結果その2(部材座標系で表された材端と中央の断面力)

これまでは、マトリックス法の理論的解説と、全体剛性行列の作成法、 Excel における入力・出力仕様について説明してきた。本節では、平面 骨組に関する解析の流れを説明する。 4.4 平面骨組応力 解析プログラム の流れ



図 4-11 平面骨組応力解析用のプログラムの流れ概要

まず、プログラムの概略的な流れについて説明する。図 4-11 には、 解析の概略的なフローチャートが示されている。このフローチャートを 見れば、解析の全般的な手続きが理解できる。解析プログラムを大きく 分けると以下のようである。

1 :使用する配列などの定義とデータ入力 2 :予備計算 3 :全体剛性行列の作成 4 :剛性行列の LDU 分解 5 :荷重ベクトルの作成

- 6:方程式を解き、変位ベクトルを求める
- 7:全体座標系における節点変位を出力する
- 8:部材座標系における部材断面力を計算し、出力する

上記の手続きを、さらに詳しく、フローチャートを用いて説明しよう。 まず、使用する配列などの定義は、次節のプログラムコードを参照され たい。変数や配列の定義が機能別に分類され、定義されている。そこで は、コメントとして名称が書かれており、使用方法は十分に理解できる ものと思う。次に、解析に必要なデータ入力は、本章で説明したとおり、

Excel のシートから行う。その順序をフローチ ャートで以下に示す。

Excel シートのデータ入力領域の設計は、既 に、前節で行っており、ここでは、プログラム コードに従って、入力順序を示す。最初に、コ ントロールデータを取得し、その値を用いて、 入力情報を格納する配列の動的確保を行う。次 に、フローチャートに書かれているように、各 情報をシートから取得する。その方法は、次節 で具体的なコードで示す。



図 4-12 データ入力の流れ

次に、予備計算であるが、本プログラムでは、以下 の3つの処理を行っている。1)境界条件から節点拘 束表を作成し、その後、未知番号の作成、2)次章で 説明するスカイライン行列のための表作成、3)部材 長の計算と全体座標系における傾き計算。これらの処 理をフローチャートで示す。



クリアする。その後、全 部材について同様の処理 を行い、順序良く各部材 の剛性を全体剛性行列の 中に足しこんでいく。そ の手法の基本的な処理方 法は、前章で示したが、 本プログラムではより効 率的なスカイライン法を 用いるため、次章で詳細 に説明する。この部分の プログラムの流れを次の フローチャートに示す。 最初に、部材座標系に おける剛性行列を作成、 次に、その部材の座標変

をゼロ

ゼロクリアと は、配列要素の 全てを0にする ことであり、プ ログラム言語で は良く用いられ る用語である。

次に、全体剛性行列の作成法について説明する。ま ず、スカイライン行列の配列を動的確保し、その要素



図 4-14 全体剛性行列の作成フローチャート

換行列を作成する。この2つの行列を使用して、行列の3重積を行い、 全体座標系に部材剛性行列を変換する。最後に、この剛性行列を全体剛 性行列に足しこむことになる。全体剛性行列は1次元配列のスカイライ ン行列であることに注意されたい。この方法については、これも次章で 説明する。この一連の処理を、全ての部材について行い、全体剛性行列 を作成する。 次に、構築した全体剛性行列を LDU 分解し、荷重ベクトルを作成する。

ここでも、全体剛性行列 がスカイライン行列であ ることに注意して分解す る。この分解で、全体剛 性行列が特異行列である ことが判明すると、「特 異行列であり、そのため 骨組が不安定である」こ とを告げ、以後の処理を 中止する。

全体剛性行列の LDU 分 解が正常終了すると、次 に、荷重ベクトルを作成 する。まず、未知数分の 荷重ベクトルをゼロクリ アし、2 つの荷重がある かどうかをチェックする。 このチェックには、コン トロール情報の荷重デー タを用いる。

本プログラムでは、部 材荷重は等分布荷重のみ 扱っており、部材座標系 で入力する仕様となって いる。まず、基本応力で ある部材両端の荷重を求 め、次に、座標変換行列 を作成して、材端荷重ベ クトルを全体座標系の荷



図 4-15 全体剛性行列の LDU 分解と荷重ペクトルの作成

重ベクトルに変換する。その後、未知番号表を利用して、全体荷重ベクトルに足しこむ。

節点集中荷重は、全体座標系で設定する仕様となっているため、その まま、未知番号表を頼りに、全体荷重ベクトルに足しこむことになる。 これで、釣合式の右辺項である荷重ベクトルの作成が終了する。 次の処理は、連立方程式を解くことであり、既に、係数行列(全体剛 性行列)は LDU 分解されている。この分解された下 三角行列を使用して方程式を解き、変位を求める。 この手法もスカイライン法を用いているため、特殊 な方法となる。次章で詳しく説明しよう。

連立方程式を解いた得た全体座標系の変位を、未 知番号表を利用して節点毎に整理し、出力する。

最後の処理として、各部材の断面力を計算し、出 力する。本プログラムでは、部材両端の材端力を計 算し、断面力の代わりとする。材端での力の釣合か

ら、各部材の断面力は、この材端力と同一となる。また、部材中央の曲 げモーメントは、部材荷重の基本応力を参照して計算する。部材の断面 力は部材座標系で表されるため、前処理で求めた全体座標系の変位を部 材座標系に変換する必要がある。これらの処理の内容は、図 4-17 に示 すフローチャートを参照して説明する。



図 4-17 部材断面力の計算と出力



最初に、全体座標系 の変位から部材両端 の節点で立を考慮して、 取り加までです。して、 の満すので、して、 の方にのたいで、 たの変換する。 に変換する。

次に、部材座標系 の剛性行列を計算す る。この剛性行列と 両端変位との掛け算 を実施して、両端の 外力を求める。また、 部材荷重に対しては、 基本応力を利用して、 部材中央の曲げモーメントを計算する。計算した部材断面力を Excel の 「解析結果」シートに出力する。

以上で、平面骨組解析プログラムの処理内容を説明した。次節で、これらの流れを、実際の Excel VBA のコードで示す。

4.5 平面骨組応力

解析プログラム

本節では、主プログラムの全コードを示し、その内容を説明する。こ のプログラムは、解析全体の主プログラムであり、全ての配列などの定 義文を含み、「解析開始」ボタンをクリックすることによって起動され る。主プログラムの全コードを以下に示す。最初の部分は、解析プログ ラムで使用される変数や配列を定義している。使用されている変数や配 列の意味は、コメント行や定義文の後のコメントを参照されたい。使用 する配列は、固定して使用する場合を除き、全て動的確保を行っている。 動的確保する配列の大きさは、プログラムコードを見れば理解できよう。 また、主プログラムの後に、3つの配列のゼロクリアサブルーチンを示 す。このサブルーチンは、主プログラムで呼ばれている。また、主プロ グラムで呼ばれている各種のサブルーチンは既に説明したか、もしくは、 これ以後の章で説明する。いずれにしても、このテキストでは、全ての コードを示し、説明する。

最初に、平面骨組で使用する変数と配列について説明する。開発する 平面骨組のプログラムで使用する配列は、全て動的に領域確保を行う。 これは、他の言語でも同様で、使用する配列は、実行中に必要となるメ モリー量を計算し、確保する必要がある。VBA では、この動的な領域確 保は非常に簡単で、次のようなコードで行われる。例えば、スカイライ ン行列を保存する領域は、次のように仮の配列定義を行い、

Dim gsky() As Double 'スカイライン剛性行列(1次元配列で使用)

その後、処理が実際に進む中でその領域の大きさを計算し、次のコード を用いて再設定する。この操作で必要最小限の領域が確保されたことに なる。

ReDim gsky(n_skyline)

以下に、平面骨組解析用プログラムで使用する代表的な変数と配列を 示す。無論、読者は、この変数名を使用する必要はないが、名前を付け る場合は、具体的にその内容が分かることが大切である。ここで示す変 数と配列の定義は、主プログラム、つまり、「解析開始」 ボタンがクリ ックされたとき、最初に動作するプログラムで定義する。

Private Sub CommandButton1_Click() *_____ 平面骨組プログラム(スカイライン法) *_____ '----- ワーク変数として使用 Dim i As Integer Dim j As Integer Dim k As Integer Dim F_cel Dim i_free As Integer Dim j_free As Integer Dim memb As Integer Dim iexit As Integer Dim istable As Integer Dim mx As Integer '----- コントロール情報 Dim n_point As Integer '節点数 Dim n Member As Integer '部材数 Dim n_Element As Integer '要素数 Dim n_kyokai As Integer '境界節点数 Dim n_M_load As Integer '部材荷重数 Dim n_P_load As Integer '節点荷重数 '----- 入力情報用 Dim Point() As Double '節点座標 Dim Member() As Integer '部材情報 Dim Element() As Double '断面情報 Dim Kyokai() As Integer '境界条件 Dim Load_M() As Double '部材荷重 Dim load_P() As Double '節点荷重 '------ 拘束表 Dim F_rest() As Integer '節点拘束(未知番号)表 Dim nsum_d() As Integer 'スカイライン行列の対角項番号 Dim n_free As Integer '自由度数 Dim n_max_sky As Integer 'スカイライン行列の最大数 '----- 解析用 Dim gsky() As Double 'スカイライン剛性行列(1次元配列で使用) Dim gsky_d() As Double 'LDU 分解後の対角項 Dim disp() As Double '変位ベクトル Dim pload() As Double '荷重ベクトル Dim nwork() As Integer 'ワーク領域(LDU分解で使用) Dim twork() As Double 'ワーク領域(LDU分解で使用) Dim al() As Double '部材長 Dim sin_cos() As Double '部材の傾き(sin とcos) Dim C_M_Q() As Double '部材荷重の基本応力 '----- ワーク領域として使用 Dim ak(6, 6) As Double '部材座標系の部材剛性行列 Dim akk(6, 6) As Double '全体座標系の部材剛性行列 Dim R(6, 6) As Double '部材回転行列 1:コントロールデータ入力 •_____

```
n_point = Range("B3").Value
n_Member = Range("B4").Value
n_Element = Range("B5").Value
n_kyokai = Range("B6").Value
n_M_load = Range("B7").Value
n_P_load = Range("B8").Value
•_____
     2:解析に必要な配列の動的確保
۰_____
'------ 入力情報用
ReDim Point(2, n_point)
ReDim Member(3, n_Member)
ReDim Element(3, n_Element)
ReDim Kyokai(4, n_kyokai)
ReDim al(n_Member)
ReDim sin_cos(2, n_Member)
ReDim C_M_Q(3, n_Member)
If (n_M_load <> 0) Then
ReDim Load_M(4, n_M_load)
End If
If (n_P_load <> 0) Then
ReDim load_P(4, n_P_load)
End If
'------ 拘束表
ReDim F_rest(3, n_point)
'_____
     3:解析情報の入力
۰<u>ــــ</u>
'------1.節点座標
F_cel = "E2"
For i = 1 To n_point
Point(1, i) = Range(F_cel).Offset(i, 0)
Point(2, i) = Range(F_cel).Offset(i, 1)
Next
'------2.部材情報
F_cel = "12"
For i = 1 To n_Member
Member(1, i) = Range(F_cel).Offset(i, 0)
Member(2, i) = Range(F_cel).Offset(i, 1)
Member(3, i) = Range(F_cel).Offset(i, 2)
Next
'------3.要素情報
F cel = "N2"
For i = 1 To n_Element
Element(1, i) = Range(F_cel).Offset(i, 0)
Element(2, i) = Range(F_cel).Offset(i, 1)
Element(3, i) = Range(F_cel).Offset(i, 2)
Next
'------ 4.境界情報
F_cel = "R2"
For i = 1 To n_kyokai
Kyokai(1, i) = Range(F_cel).Offset(i, 0)
Kyokai(2, i) = Range(F_cel).Offset(i, 1)
Kyokai(3, i) = Range(F_cel).Offset(i, 2)
```

```
Kyokai(4, i) = Range(F_cel).Offset(i, 3)
Next
'------5.節点荷重
If (n_P_load <> 0) Then
F_cel = "W2"
For i = 1 To n_P_load
load_P(1, i) = Range(F_cel).0ffset(i, 0) ' 節点番号で整数であるがここでは実数として保存
load_P(2, i) = Range(F_cel).Offset(i, 1)
load_P(3, i) = Range(F_cel).Offset(i, 2)
load_P(4, i) = Range(F_cel).Offset(i, 3)
Next
End If
'-----6.部材荷重
If (n_M_load <> 0) Then
F_cel = "AB2"
For i = 1 To n_M_load
Load_M(1, i) = Range(F_cel).Offset(i, 0) '部材番号で整数であるがここでは実数として保存
Load_M(2, i) = Range(F_cel).Offset(i, 1)
Next
End If
·-----
Range("A18").Value = "データ入力終了"
4:予備計算1(節点拘束表から未知番号に変換)
۰_____
 Call Cal_Yobi(F_rest, n_point, n_kyokai, Kyokai, n_free)
 Range("B20") = n_free
•_____
      予備計算2(スカイライン用対角項番号表作成)
·_____
'-----表領域の動的確保
ReDim nsum_d(n_free)
Call Set_sky_table(n_free, F_rest, nsum_d, n_Member, Member, n_skyline)
Range("B21") = n_skyline
۰_____
     予備計算3(部材長さとsinとcosを計算)
'-----
 Call Al_member(n_Member, Member, Point, al, sin_cos)
 Range("A18").Value = "予備計算終了"
5:全体剛性行列作成
۰_____
'------ 解析用動的領域確報
ReDim gsky(n_skyline)
ReDim gsky_d(n_free)
ReDim disp(n_free)
ReDim pload(n_free)
ReDim nwork(n_free)
ReDim twork(n_free)
'------ スカイライン行列をゼロセット
Call Set_sky_zero(n_skyline, gsky)
'------ 全体剛性行列作成
For memb = 1 To n_{Member}
```

```
mx = Member(3, memb)
 Call Cal_k(memb, ak, al, mx, Element)
 Call Cal_rot(memb, R, sin_cos)
 Call Cal_all_rot(ak, akk, R)
 Call Build_sky_k(memb, gsky, akk, F_rest, nsum_d, Member)
Next
'------ 全体剛性行列 L D U 分解
 Call Decomp_sky(n_free, nsum_d, gsky, gsky_d, nwork, twork, iexit, istable)
 Range("B22") = iexit
 Range("B23") = istable
 If (iexit <> 0) Then GoTo Err_C
 Range("A18").Value = "剛性計算終了"
'_____
   6:荷重ベクトルの作成
•
'----- 荷重ベクトルのゼロクリア
Call zero_pload(n_free, pload)
'------ 基本応力項のゼロクリア
Call zero_CMQ(n_Member, C_M_Q)
'----- 部材分布荷重
 If (n_M_load <> 0) Then
  Call Set_load_M(pload, n_M_load, Load_M, C_M_Q, al, Member, F_rest, sin_cos)
End If
'------ 節点荷重
If (n_P_load <> 0) Then
 Call Set_load_S(pload, n_P_load, load_P, F_rest)
End If
'_____
Range("A18").Value = "荷重計算終了"
'_____
    7: 釣合式を解く
۰_____
Call solv_sky(n_free, nsum_d, gsky, gsky_d, nwork, twork, pload, disp)
۰_____
    8:全体変位出力
·____
Call Out_disp(disp, n_point, F_rest)
'-----
Range("A18").Value = "解法計算終了"
'_____
   9:部材応力の計算と出力
۰<u>ــــ</u>
Call Cal_stress(n_Member, Member, al, sin_cos, disp, F_rest, Element, C_M_Q)
۰_____
Range("A18").Value = "応力計算終了"
'_____
   解析終了
۰_____
GoTo End_C
Err_C:
Range("A18").Value = "分解エラー"
End C:
End Sub
```

```
' スカイライン剛性のゼロクリア
'
Private Sub Set_sky_zero(n_skyline, gsky)
Dim i As Integer
For i = 1 To n_skyline
gsky(i) = 0#
Next
End Sub
```

```
'

荷重ベクトルのゼロクリア

Private Sub zero_pload(n_free, pload)

Dim i As Integer

For i = 1 To n_free

pload(i) = 0#

Next

End Sub
```

```
'基本応力項のゼロクリア
'
Private Sub zero_CMQ(n_Member, C_M_Q)
Dim i As Integer
Dim j As Integer
For i = 1 To n_Member
For j = 1 To 3
C_M_Q(j, i) = 0#
Next
Next
End Sub
```

4.6 課題

本章では、平面骨組に必要となる最低限の入力情報をまとめ、Excel シート上にデザインした。この入力仕様に従って、図 4-1 から 4-10 ま でを参考にして、Excel シート上に入力データ領域及び出力データ領域 を設定する。また、ボタンを押すと起動する主プログラムが紹介されて いるので、このコードをこれまでの平面骨組プログラムの中に組み込む。 その際、添付されている 3 つのサブルーチンも同時に書き込みを行う。 以上を間違いなく実施し、プログラムを作成されたい。

これで、平面骨組の解析プログラムは一部動作するようになったはず である。特に、入力データの読み込み部分は動作するはずである。そこ で、デバック操作を始める前に、入力データが正常に読み込まれている かどうか、直接セル上に出力して、確認されたい。

4.7 まとめ

本章では、平面骨組に必要となる入力情報をまとめ、Excelのシート 上に設計した。また、骨組解析プログラムの全体処理についてフローチ ャートを用いて説明し、その処理の流れと内容を理解した。また、プロ グラムの続きとして、ボタンで起動する主プログラムを組み込み、デバ ック作業に入った。